
The “Harmony Lemma”, as formulated by Sangiorgi & Walker, establishes the equivalence between
the labelled transition semantics and the reduction semantics in the 7-calculus. Despite being a
widely known and accepted result for the standard mw-calculus, this assertion has never been rigor-
ously proven, formally or informally. Hence, its validity may not be immediately apparent when
considering extensions of the m-calculus. Contributing to the second challenge of the Concurrent
Calculi Formalization Benchmark — a set of challenges tackling the main issues related to the mech-
anization of concurrent systems — we present a formalization of this result for the fragment of the
m-calculus examined in the Benchmark. Our formalization is implemented in Beluga and draws
inspiration from the HOAS formalization of the LTS semantics popularized by Honsell et al. In pass-
ing, we introduce a couple of useful encoding techniques for handling telescopes and lexicographic
induction.

1 Introduction

At page 51 of their “bible” on the m-calculus [19], Sangiorgi & Walker state the Harmony Lemma,
regarding the relationship between the reduction semantics and the transitional one (LTS). The sketch of
the proof starts as follows:

Rather than giving the whole (long) proof, we explain the strategy and invite the reader to
check some of the details [. .. ]

While this informal style of proof, akin to the infamous “proof on a napkin” championed by de Millo
and colleague, may be suitable for a (long) textbook, it might not be applicable to emerging calculi
with more unconventional operational semantics. Although the theorem is undisputed within the well-
established framework of the m-calculus, this assurance may not extend to these developing calculi.
In such instances, a more rigorous approach, potentially in the form of a machine-checked proof, is
advisable.

These considerations are of course not novel: they have been prominently argued for in the POPL-
Mark challenge [2]] and subsequent follow-ups [1,[7]. The recent Concurrent Calculi Formalization
Benchmark [4]] (CCFB in brief) introduces a new collection of benchmarks addressing challenges en-
countered during the mechanization of models of concurrent and distributed programming languages,
with an emphasis on process calculi. As with POPLMark, the idea is to explore the state of the art in the
formalization in this subarea, finding the best practices to address their typical issues and improving the
tools for their mechanization.

CCFB considers in isolation three aspects that may be problematic when mechanizing concurrency
theory: linearity, scope extrusion, and coinductive reasoning. Scope extrusion is, of course, the method
by which a process can transfer restricted names to another process, as long as the restriction can be

1“Social Processes and Proofs of Theorems and Programs”, CACM 22-5, 1979.

2 Formalizing the Harmony Lemma in Beluga

safely expanded to include the receiving process. This phenomenon has been captured in two different,
yet equivalent ways of formulating the operational semantics of the 7-calculus:

1. a reduction system, which avoids explicit reasoning about scope extrusion by using structural
congruence;

2. alabelled transition system, which introduces a new kind of action to handle extrusion directly: in
doing so, it breaks shared conventions such as a-equivalence

The second challenge in the Concurrent Calculi Formalization Benchmark (CCFB.2) consists in mecha-
nizing these two operational semantics and relating them via the aforementioned Harmony Lemma.

Obviously, we are not the first to address the mechanization of the w-calculus (although we seem to
be the first to tackle the Harmony result): given the challenges that it poses (various kind of binders with
somewhat unusual properties compared to the A-calculus), there is a long tradition starting with [[11] and
mostly developed with encodings based on first-order syntax such as de Brujin indexes — see [4] for a
short review of the literature w.r.t. scope extrusion. As often remarked, concrete encodings will get you
there, but not effortlessly: an estimation of 75% of the development being devoted to the infrastructure
of names handling is not uncommon [8]]:

“Technical work, however, still represents the biggest part of our implementation, mainly
due to the managing of De Bruijn indexes [...] Of our 800 proved lemmas, about 600 are
concerned with operators on free names.”

It is not surprising that specifications based on higher-order abstract syntax (HOAS) soon emerged,
first only as animations, see [12] in AProlog and [9] in LF. Moving to meta-reasoning, we can roughly
distinguish two main approaches:

1. “squeezing” HOAS into a general proof assistant: there is a plethora of approaches, but w.r.t the

m-calculus this has been investigated by Despeyroux [6] and then systematically by Honsell and
his colleagues, starting with [[10] and then addressing other calculi;

2. the Pfenning-Miller “two-level approach” of separating the specification from the reasoning logic,
whose culmination, as far as the m-calculus is concerned, is the most elegant version presented
in [21]] and later implemented in Abella.

We fall in the second camp and we offer a Beluga [[18] mechanization of CCFB.2 together with a
detailed informal proof, filling all the gaps left by the quoted sketch. Along the way, we introduce (or
simply rediscover) a couple of Beluga tricks to encode telescopes (i.e. n-ary sequences of binders) and to
simulate lexicographic induction. We also prove another folk result, namely the equivalence between the
early and late LTS, as well as what is sometimes called “internal adequacy” [[10], that is the equivalence
between the LTS encoding from the Honsell paper with the one in [21].

Informal and formal proofs in all their glory can be found here [5]. In the text, the statements of
informal lemmas and theorems are hyperlinked to their formalization in the repository. For reasons of
space, we will assume familiarity with the basic notions of the w-calculus as in [15]], as well as a working
knowledge of Beluga, both of its syntax and more importantly of its approach to proof checking.

2 The w-Calculus and its Operational Semantics

In this section, we quickly recall the main notions involved, so as to make the paper self-contained. For
more details see [[19]].

2There are also intermediate approaches that save o-equivalence, such as Parrow’s LTS with structural congruence [15] or
Milner’s notion of abstraction and concretion as formalized for example in [3].

2.1 Syntax

We assume the existence of a countably infinite set of names, ranged over by x,y,... We make no other
assumption about names, since the syntax of processes in CCFB.2 does not consider (mis)match. In fact,
to concentrate in isolation on scope extrusion, sums and replications are ignored as well:

PQ == 0| x(y).P | xyP | (P|Q) | (vx)P

The input prefix x(y).P and the restriction (vy)P both bind the name y in P. Any other occurrence of
names in a process is free. The sets of free and bound names occurring in a process (fn(P) and bn(P)
respectively) are defined as usual.

In the mathematical presentation of the operational semantics, we adopt the following slightly weaker
variable conventior: 1) given a process, it is possible to o.-rename the bound occurrences of variables
within it; 2) the bound names of any processes or actions under consideration can be chosen different
from the names occurring free in any other entities under consideration.

2.2 Reduction Semantics

We define structural congruence (=) and reduction (—) as the smallest binary relations over processes,
respectively satisfying the axioms in Fig.[Il The notation Q{y/z} represents capture-avoiding substi-
tution of y for z in the process Q. Note that we have chosen to present congruence as the compatible
refinement of the six basic axioms, rather than using process contexts as in [19], since the latter tend to
be problematic w.r.t. a HOAS formalization.

2.3 Labelled Transition System Semantics

The syntax of actions is the following:

a:=x(y) | Xy | x(y) |t

In the input action x(y) and in the bound output action %(y), the name x is free and y is bound; in the free
output action Xy, both x and y are free. The sets of free names, bound names and names occurring in an
action (bn(), fn(a) and n(a) respectively) are defined accordingly. The transition relation - — - is the
smallest relation which satisfies the rules in Fig.

Unlike the reduction semantics, the transitional semantics directly addresses scope extrusion via the
two S-CLOSE rules in interaction with S-OPEN: recall how the former rules are not closed under -
conversion, since the bound name z must occur free in the other premise.

The LTS introduced here is the late semantics, as opposed to the early one adopted by the Benchmark.
However, as remarked in [15]], “it is a matter of taste which semantics to adopt”. We indeed prove this
equivalence in Appendix [Al

2.4 The Harmony Lemma
In [19], the Harmony Lemma reads as:

i p=% Q implies P Z= 0.

3Variable conventions are used in a rather loose way in the literature, e.g. Parrow and Sangiorgi & Walker adopt the same
convention, but end up with different provisos in the operational semantics rules.

PAR-ASSOC PAR-UNIT PAR-COMM
Pl(Q|R)=(P[Q)|R PlO=P Plo=0|P
Sc-EXT-PAR
Sc-EXT-ZERO x & fn(Q) Sc-EXT-RES
(vx)0 =0 (V)P |Q = (vx)(P|Q)  (vx)(vy)P = (vy)(vx)P
C-IN C-Out C-PAR C-RES
P=0 P=0 P=F P=Q
x(y).P = x(y).0 xy.P = xy.0 P|Q= P |0 (vx)P = (vx)Q
C-Sym C-TRANS
C-REF P=0 P=Q Q=R
P=P 0=P P=R
R-PAR
R-CoMm P—=Q
xy.P|x(z).0 — P|Q{y/z} PIR— Q|R
R-RES R-STRUCT
P—Q P=P P =0 0=0
(vx)P — (vx)Q P—Q

Figure 1: Congruence and reduction rules.

ii. P5=QiffP— Q.
The juxtaposition of symbols denotes relational composition (e.g. P =% Qdenotes P=Rand R % Q
for some R). The first assertion is a direct consequence of Lemma as detailed at page [6] which is

instrumental to prove the right-to-left direction of the equivalence result. The latter breaks down into the
following theorems:

1. Every transition through a 7 action corresponds to a reduction;

2. Given a reduction of P to Q, P is able to make a T-transition to some Q' congruent to Q.
In the interest of setting the stage for anybody who wishes to give a solution to CCFB.2, we start by stat-
ing a few technical lemmas about substitutions that are used in both directions of the Harmony Lemma,
while being often left unsaid.
Lemma S1 Q{x/x} = Q.
Lemma S2 [fx ¢ fn(Q), then Q{y/x} = Q.
These two lemmas are proved by induction on the structure of the process Q. A consequence of the latter
is the following: if x ¢ fn(Q), then P{y/x} | 0 = (P | Q){y/x}.

Finally, we state a stability result for structural congruence under substitutions, only used in the
second direction of Harmony:

Lemma S3 If P = Q, then P{y/x} = Q{y/x}.
This lemma is proved by induction on the structure of the given derivation.

S-IN S-OuTt
x(z).P @, p Xy.P Dop
S-PAR-L S-PAR-R
PSP bn(a)Nfn(Q)=0 050  bn(a)Nnfn(P)=0
PIQ % PO PlQ % PO
S-CoMm-L S-CoM-R
PP 0y PP 9%y
PlQ 5 P Q{y/} PlQ 5 Py/zt|Q
S-RES ~ S-OPEN
PEP  z¢n(a) PSP z#x
(va)P % (v2)P' (va)p 2, pr
i S-CLOSE-L S-CLOSE-R_
Py g Py o™y
PlQ = (vo)(P'| Q) PlQ = (va)(P'| Q)

Figure 2: Transition rules.

2.4.1 Theorem[I: 7-Transition Implies Reduction

The proof of the first direction relies on three key lemmas which describe rewriting (up to structural
congruence) of processes involved in input and output transitions.

7|Lemma 1.1 If Q M Q' then there exist a finite (possibly empty) set of names wi,...,w, (with

x,y # w; Vi = 1,...,n) and two processes R,S such that Q = (vwy)...(vw,)(x(y).R | S) and
Q' =(vwy)...(Vvw,) (R S).

7|Lemma 1.2 If Q 2, Q' then there exist a finite (possibly empty) set of names wy,...,w, (with
x,y #w; Vi = 1,...,n) and two processes R,S such that Q = (vwy)...(vw,)(Xy.R | S) and
Q' =(vwy)...(Vvw,) (R S).

Z|Lemma 1.3 If Q &) Q' then there exist a finite (possibly empty) set of names wi,...,wy,
(with x & {z,w1,...,w,}) and two processes R,S such that Q = (vz)(vwy)...(vw,)(Xz.R | S) and
Q' =(vwy)...(Vvw,) (R S).

These three lemmas are proved by induction over the structure of the given transition. We observe
that the presence of a sequence of binders is not an issue in the informal presentation; on the other hand,
from the mechanization point of view, these sequences are challenging to encode in a framework where
the meta-level binder is unary.

Theorem 1 P 5 Q implies P — Q.


https://github.com/GabrieleCecilia/concurrent-benchmark-solution/blob/main/code/2_input_rewriting.bel
https://github.com/GabrieleCecilia/concurrent-benchmark-solution/blob/main/code/3_free_output_rewriting.bel
https://github.com/GabrieleCecilia/concurrent-benchmark-solution/blob/main/code/4_bound_output_rewriting.bel
https://github.com/GabrieleCecilia/concurrent-benchmark-solution/blob/main/code/5_theorem1.bel

6 Formalizing the Harmony Lemma in Beluga

The theorem is proved by induction on the structure of the given transition. If the latter consists
of an explicit interaction of processes in a parallel composition, we apply the aforementioned lemmas
to rewrite processes involved in specific transitions up to congruence; we then construct the desired
reduction through a chain of congruence and reduction rules.

Corollary 1.1 P 5= Q entails P — Q.

2.4.2 Theorem 2: Reduction Implies t-Transition

The other direction starts with five technical lemmas regarding free and bound names in specific tran-
sitions. They are instrumental, together with the variable convention, to the firing of the appropriate
transitions.

Lemma 2.1 IfP 5 P/, then x,y € fn(P).
Lemma 2.2 IfP 0), P!, then x € fn(P).

Lemma 2.3 If P @), P', then x € fn(P) and z € bn(P).
Lemma 24 IfP 5 P, x ¢ n(a) and x ¢ fn(P), then x & fa(P').
Lemma 2.5 I[fP =P/, then x € fn(P) < x € fn(P').

The first four lemmas follow by induction over the structure of the given transition. The last by
induction on the congruence judgment.
The next key ingredient is establishing that structural congruence is a strong late bisimulation.

Lemma 2.6 Ler P = Q.

1. If P2 P', then there exists a process Q' such that Q 2 Q' and P' = Q.
2. IfQ 2 O, then there exists a process P' such that P25 P' and P' = Q.

The two statements need to be proven at the same time by mutual induction over the derivation of the
congruence judgment and case analysis on the given transition.

Finally, a rewriting lemma for reduction, again proven by induction on the structure of the given
reduction judgment:

Lemma 2.7 If P — Q then there exist three names x,y and z, a finite (possibly empty) set of names
Wi,...,wy, and three processes Ry,R, and S such that P = (vwy)...(vw,)((Xy.R; | x(2).R2) | S) and

Q= (vwi1)...(vwa)((R1 | Ro{y/2}) | ).
Theorem 2 P — Q implies the existence of a Q' such that P Q' and Q = Q.

The proof follows immediately from the application of Lemmas [2.6]and 2.71

3 Beluga Formalization

This section provides an overview of the formalization of the definitions and proofs introduced in the
previous section with the proof assistant Beluga. The complete formalization is accessible at [5]].


https://github.com/GabrieleCecilia/concurrent-benchmark-solution/blob/main/code/6_stepcong_lemma.bel
https://github.com/GabrieleCecilia/concurrent-benchmark-solution/blob/main/code/7_reduction_rewriting.bel
https://github.com/GabrieleCecilia/concurrent-benchmark-solution/blob/main/code/8_theorem2.bel

3.1 Syntax

Fig.[Bldisplays the syntax of names and processes. Since names are just an infinite set, we encode them
with an LF type names without any constructor, which will be extended with new inhabitants dynamically
in the operational semantics. This is made possible by the declaration

schema ctx = names;

indicating that all relevant judgments involving open terms (processes) are formulated in contexts cate-
gorized by the above schema.

Processes are encoded using (weak) HOAS, as well trodden in the literature: input and restrictions
abstract over names, in particular the input process x(y).P is encoded by the term p_in X \y. (P y),
where the bound name y in P is represented by the implicit argument of the LF function \y. (P y).
As usual, a-renaming and capture-avoiding substitutions are automatically implemented by the meta-
language. From now on, semicolons and infix constructor declarations will be omitted from code snippets
for brevity.

LF names: type =

LF proc: type =
| p_zero: proc
| p_in: names — (names — proc) — proc
| p_out: names — names — proc — proc
| p_par: proc — proc — proc
| p_res: (names — proc) — proc

Figure 3: Encoding of names and processes.

Recall that LF types are not inductive. While the LF type names has no constructor, the contextual
type [g b procl, for g € ctx, denotes the set of open processes built out of LF variables ranging over
names.

3.2 Reduction Semantics

Congruence and reduction are encoded by the type families cong and red respectively, as presented in
Fig.[dl As usual, we use universal quantification {x:names} to descend into binders, e.g. in the compati-
bility rule for restriction. Scope extension is realized in rule sc_ext_par by simply not having Q depend
on the restricted channel, hence meeting the side condition x ¢ fn(Q) in the Sc-EXT-PAR automatically.
In the same vein, substitution in rule R-CoM is encoded by meta-level -reduction.

3.3 Labelled Transition System Semantics

We follow Honsell et al. [[10] for the encoding of the late LTS semantics. We declare two different
relations for transitions via free and bound actions. The result of a free transition is a process, while the
result of a bound transition is a process abstraction: instead of explicitly stating the bound name involved
in the transition, that name is the argument of the aforementioned function. This is reflected by the
encoding of free and bound actions, which only mention the free names involved. Fig.[5shows the types

% Structural Congruence
LF cong: proc — proc — type =
% Abelian Monoid Laws for Parallel Composition
| par_assoc: cong (P p_par (Q p_par R)) ((P p_par Q) p_par R)
| par_unit: cong (P p_par p_zero) P
| par_comm: cong (P p_par Q) (Q p_par P)
% Scope Extension Laws
| sc_ext_zero: cong (p_res (\x.p_zero)) p_zero
| sc_ext_par: cong ((p_res P) p_par Q) (p_res (\x.((P x) p_par Q)))
| sc_ext_res: cong (p_res \x.(p_res \y.(P x y))) (p_res \y.(p_res \x.(P x y)))
% Compatibility Laws
| c_in: ({y:names} cong (P y) (Q y)) — cong (p_in X P) (p_in X Q)
| c_out: cong P Q — cong (p_out X Y P) (p_out X Y Q)
| c_par: cong P P’ — cong (P p_par Q) (P’ p_par Q)
| c_res: ({x:names} cong (P x) (Q x)) — cong (p_res P) (p_res Q)
% Equivalence Relation Laws
| c_ref: cong P P
| c_sym: cong P Q — cong Q P
| c_trans: cong P Q — cong Q R — cong P R

% Reduction
LF red: proc — proc — type =
| r_com: red ((p_out X Y P) p_par (p_in X Q)) (P p_par (Q Y))
| r_par: red P Q — red (P p_par R) (Q p_par R)
| r_res: ({x:names} red (P x) (Q x)) — red (p_res P) (p_res Q)
| r_str: P cong P’ — red P’ Q° — Q’> cong Q — red P Q

Figure 4: Encoding of congruence and reduction.

f_act and b_act encoding free and bound actions and the two mutually defined type families fstep
and bstep which encode free and bound transitions respectively@. Note that none of the side conditions
of the transition rules need to be explicitly stated, nor do we need the axioms and additional freshness
predicates as in [10].

HOAS encodings customarily come with an (informal) adequacy proof, ensuring that there is a com-
positional bijection between the mathematical model and its encoding (in canonical form). While this is
fairly obvious for processes, congruence and reduction, it is less so w.r.t. the LTS semantics. Luckily,
this has been carefully proven both in [10] and in [21]] for a related version. We refer to those papers for
further details and to the repository for a Beluga proof of the “internal” adequacy of those two encodings.

3.4 The Harmony Lemma

One of the expected yet very much appreciated payoffs of a HOAS encoding is that most (in fact all but
Lemma [2.4) boilerplate lemmas about names, occurrences and substitution vanish. We are referring to
the substitution Lemmas [STS3] as well as the free/bound names Lemmas and

“Interestingly, a similar approach is pursued by Cheney in his nominal encoding in aProlog, see
https://homepages.inf.ed.ac.uk/jcheney/programs/aprolog/examples/picalc.apl



https://homepages.inf.ed.ac.uk/jcheney/programs/aprolog/examples/picalc.apl

% Free Actions % Bound Actions

LF f_act: type = LF b_act: type =
| f_tau: f_act | b_in: names — b_act
| f_out: names — names — f_act | b_out: names — b_act

% Transition Relation
LF fstep: proc — f_act — proc — type =
fs_out: fstep (p_out X Y P) (f_out X Y) P
fs_parl: fstep P A P’ — fstep (P p_par Q) A (P’ p_par Q)
fs_par2: fstep Q A Q> — fstep (P p_par Q) A (P p_par Q°)
fs_coml: fstep P (f_out X Y) P’ — bstep Q (b_in X) Q’
— fstep (P p_par Q) f_tau (P’ p_par (Q’ Y))
| fs_com2: bstep P (b_in X) P> — fstep Q (f_out X Y) Q’

— fstep (P p_par Q) f_tau ((P’ Y) p_par Q’)
| fs_res: ({z:names} fstep (P z) A (P’ z)) — fstep (p_res P) A (p_res P’)
| fs_closel: bstep P (b_out X) P’ — bstep Q (b_in X) Q’

— fstep (P p_par Q) f_tau (p_res \z.((P’ z) p_par (Q’ z)))
| fs_close2: bstep P (b_in X) P> — bstep Q (b_out X) Q’

— fstep (P p_par Q) f_tau (p_res \z.((P’ z) p_par (Q’ z)))

and bstep: proc — b_act — (names — proc) — type =

bs_in: bstep (p_in X P) (b_in X) P

bs_parl: bstep P A P> — bstep (P p_par Q) A \x.((P’ x) p_par Q)

bs_par2: bstep Q A Q° — bstep (P p_par Q) A \x.(P p_par (Q’ x))

bs_res: ({z:names} bstep (P z) A (P’ 2z))

— bstep (p_res P) A \x.(p_res \z.(P’ z x))

| bs_open: ({z:names} fstep (P z) (f_out X z) (P’ z)) — bstep (p_res P)(b_out X) P’

Figure 5: Encoding of actions and transition.

3.4.1 Theorem [Tt 7-Transition Implies Reduction

We start with the encoding of Lemmas and There are two issues, one standard, the other
slightly more challenging. For one, the conclusion of these lemmas includes an existential quantification,
and Beluga lacks such a construct; the usual workaround consists of defining a new type family which
encodes the existential quantification. Secondly, and more seriously, the statements refer to sequences
of binders (here restrictions), sometimes referred to as felescopes. We can give a combined solution to
these items by encoding them inductively, where the base case describes when the existential holds for
the empty sequence and the inductive one adds one more binder. More specifically for Lemma [I.1l we
say that the congruences (x) (Q = (vwy)...(vw,)(x(y).R | S) and Q' = (vwy)...(vw,)(R|S)) hold for
two processes Q and Q' iff one of the following holds:

i. 0=x(y).R|Sand Q' =R|S;
ii. Q= (vw)P, Q' = (vw)P and the congruences (x) hold for P and P'.

We list the definition of the type family ex_inp_rew that encodes the above judgment; the types
ex_fout_rew and ex_bout_rew are defined analogously.

LF ex_inp_rew: proc — names — (names — proc) — type =
| inp_base: Q cong ((p_in X R) p_par S) — ({y:names} (Q’ y) cong ((R y) p_par S))
— ex_inp_rew Q X Q’




10 Formalizing the Harmony Lemma in Beluga

rec bs_in_rew: (g:ctx) [g F bstep Q (b_in X) \y.Q’[..,yl]
— [g F ex_inp_rew Q X \y.Q’[..,yl] =
/ total b (bs_in_rew _ _ _ _ b) /
fn b = case b of
| [gF bs_in] = [g I inp_base (c_sym par_unit) \y.(c_sym par_unit)]
| [g F bs_parl B1]:[g  bstep (P p_par R) (b_in X) (\y.(P’ p_par (R[..1)))] =
let [g F D1] = bs_in_rew [g F B1] in
let [g F D2] = bs_in_rew_parl [g - R] [g F D1] in [g F D2]
| [g F bs_par2 B2]:[g I bstep (R p_par P) (b_in X) (\y.((RI[..]) p_par P’))] =
let [g F D1] = bs_in_rew [g - B2] in
let [g + D2] = bs_in_rew_par2 [g - R] [g F D1] in [g F D2]
| [g - bs_res \y.B[..,yll =
let [g,y:names  D1[..,y]] = bs_in_rew [g,y:names ~ B[..,y]l] in
let [g F D2] = bs_in_rew_res [g,y:names - D1[..,y]] in [g - D2]

Figure 6: Proof of Lemma[L.1l

| inp_ind: Q cong (p_res P) — ({y:names} (Q’ y) cong (p_res (P’ y)))
— ({w:names} ex_inp_rew (P w) X \y.(P> y w)) — ex_inp_rew Q X Q’

We prove each lemma by defining a total recursive function which receives a contextual deriva-
tion of an input/free output/bound output transition respectively and returns a contextual object of the
corresponding existential type. We provide the proof term of the bs_in_rew function, which proves
Lemmal[L1l in Fig.[6} the functions encoding Lemmas [[.2]and [[.3] are omitted.

The proof proceeds by induction on the structure of the given assumption. If it has been obtained
through the S-IN rule, we are in the base case of the existential with no binders: hence, we conclude

immediately modulo symmetry of congruence. In case the input transition has been obtained through the

S-PAR-L rule, it can be rewritten as P | R @) P" | R, with the transition B1: P @) P’ as hypothesis. In

such situation we can apply a recursive call of the bs_in_rew function to B1, obtaining an object D1 of type
ex_inp_rew encoding the rewriting for P and P’; to conclude, we appeal to the lemma bs_in_rew_pari
in order to unfold D1 and build the required existential object encoding the rewriting for P | R and P’ | R.
Here is the signature of the above lemma:

rec bs_in_rew_parl: (g:ctx) {R:[g I procl} [g F ex_inp_rew Q X \y.Q’[..,y]]
— [g F ex_inp_rew (Q p_par R) X \y.(Q’[..,y] p_par R[..1)] = ...

The other two cases of the main proof follow the same pattern and require similar auxiliary lemmas.
We are now ready to discuss the proof of the main result of this section, namely that P 50 implies
P — Q (Theorem [I):

rec fstep_impl_red: (g:ctx) [g F fstep P f_tau Q] — [ghH P red Q] = ...

The proof proceeds by induction on the derivation of [g - fstep P f_tau Q]. Mirroring the in-
formal proof, in certain subcases it is enough to apply a recursive call of the function on a structurally
smaller 7-transition and return the desired object of type [g + P red Q].In the other subcases, we apply
the functions bs_in_rew, fs_out_rew and bs_out_rew on the given input/output/bound output transi-
tions, obtaining the corresponding objects of existential type, viz. the cited ex_inp_rew and its “siblings”
ex_fout_rew and ex_bout_rew. Then, we would like to conclude by applying some auxiliary functions
which unfold these objects and build the desired reduction. Here, we face a major hurdle, not so much
in writing down the proof terms, but in having them checked for termination, which is, after all, what

guarantees that the inductive structure of the proof is correct. Let us see one of such lemmas, which
emerges in the subcase where a S-CoM-L rule is applied:
rec fs_coml_impl_red: (g:ctx) [g F ex_fout_rew P1 X Y Q1]

— [g F ex_inp_rew P2 X \x.Q2[..,x]]
— [g + (P1 p_par P2) red (Q1 p_par Q2[..,Y]D] = ...

The proof needs to consider both hypotheses, in order to unfold the two existential types — recall that the
telescopes force upon us an inductive encoding of those existentials. In other terms, verification of the
fact that this function is decreasing would need to appeal to a form of lexicographic induction.

Termination checkers are of course incomplete and adopt strict syntactic criteria to enforce it; in
particular, Beluga’s checker currently does not support lexicographic induction. One way out is to define
the fs_coml_impl_red function so that it is decreasing on the first argument only, and relies on an
auxiliary function addressing its base case that is decreasing on the second argument. The signature of
such a function is:

rec fs_coml_impl_red_base: (g:ctx) [g F P2 cong ((p_in X \x.R[..,x]) p_par S)]
— [g,w:names + Q2[..,w] cong (R[..,w] p_par S[..1)]
— [g F ex_fout_rew P1 X Y Q1]
— [g + (P1 p_par P2) red (Q1 p_par Q2[..,Y]1D] = ...

Once these lemmas are in place, the proof of the first direction of the Harmony Lemma follows
without any further drama.

3.4.2 Theorem 2: Reduction Implies t-Transition

In the other direction, having HOAS disposed of most of the technical lemmas about names and sub-
stitutions, the workhorses are the reduction rewriting Lemma and the congruence-as-bisimilarity
Lemmal[2.6l Since the former does not introduce new ideas, we discuss it first.

Given its similarity to Lemmas Lemma is implemented with the same strategy: we
define an existential type ex_red_rew that inductively encodes the existence of the telescopes and the
two congruences stated in the conclusion of the lemma.

LF ex_red_rew: proc — proc — type =
| red_base: P cong (((p_out X Y R1) p_par (p_in X R2)) p_par S)
— Q cong ((R1 p_par (R2 Y)) p_par S) — ex_red_rew P Q
| red_ind: P cong (p_res P’) — Q cong (p_res Q’)
— ({w:names} ex_red_rew (P’ w) (Q’ w)) — ex_red_rew P Q

We then implement some auxiliary functions to unfold objects of the existential type ex_red_rew in
specific subcases, for example:

rec red_impl_red_rew_par: (g:ctx) {R:[g I procl} [g F ex_red_rew P Q]
— [g F ex_red_rew (P p_par R) (Q p_par R)] = ...

Having established those, it is straightforward to prove
rec red_impl_red_rew: (g:ctx) [ghF P red Q1 — [gF ex_red_rew P Q] = ...

by induction on the structure of the given reduction judgment.

Moving on to the proof of Lemma there is a new technicality to address. We have seen how
in a HOAS setting provisos such as “x ¢ fn(P)” are realized by P being in the scope of a meta level
abstraction that binds x, but not actually depending on x. Sometimes (see [20] for other instances),
we have to convince our proof environment of this non-dependency, which is basically the content of

Lemma[2.4] in words: “given a transition P N O where x does not occur free in P, then x does not occur
free in o and Q”. Since in Beluga judgments over open terms are encapsulated in the context where they
make sense, removing these spurious dependencies amounts to “strengthening” such a context, akin to
strengthening lemmas in type theory. The lemma more formally reads as:

If T, x : names - P 25 Q,, then there are o', Q' such that & = @', O, = Q' and T + P LN o o

Not only Beluga does not have existentials (nor conjunctions), but LF also has no built-in equal-
ity notion. However, this is easy to simulate via pattern unification over canonical forms, by defining
three type families encoding equality of processes, free actions and bound actions respectively. Process
equality is defined as:

LF egp: proc — proc — type =
| prefl: eqp P P

Since we are now dealing with a property about a LF contextual object (the initial transition), the
statement in (I) has to be encoded at the computation level as an inductive type [17]. We list this
encoding, omitting the definition of its counterpart for bound transitions ex_str_bstep.

inductive ex_str_fstep: (g:ctx) [g,x:names - fstep P[..] A Q] — ctype =
| ex_fstep: {F:[g,x:names F fstep P[..] A Ql} [g I fstep P A’ Q’]
— [g,x:names - eqf A A’[..]] — [g,x:names - eqp Q Q’[..]1]
— ex_str_fstep [g,x:names - F]

Note how non-occurrence is realized using weakening substitutions, i.e. P[. .] signals that the process P
depends only on the variables mentioned in g, excluding x.
The strengthening lemma is implemented through the two following mutually recursive functions:

rec strengthen_fstep: (g:ctx) {F:[g,x:names  fstep P[..] A Q]}
— ex_str_fstep [g,x:names - F] = ...

and rec strengthen_bstep: (g:ctx) {B:[g,x:names I bstep P[..] A \z.Q[..,x,z]1%}
— ex_str_bstep [g,x:names F B] = ...

The proof follows by a long but straightforward induction on the structure of the given transition.

To state Lemma we again need to code the existential in its conclusion; however, since the
statement involves transitions through a generic action o, we actually require two new type families: one
for free transitions and one for bound transitions.

LF ex_fstepcong: proc — proc — f_act — proc — type =
| fsc: fstep Q A Q° — P’ cong Q° — ex_fstepcong P Q A P’
LF ex_bstepcong: proc — proc — b_act — (names — proc) — type =
| bsc: bstep Q A Q° — ({x:names} (P’ x) cong (Q’ x)) — ex_bstepcong P Q A P’

The reader may wonder why the process P is mentioned in the fsc and bsc constructors, as it does not
play any role: indeed, it is a trick to please Beluga’s coverage checker when this definition is unfolded in
the rest of the development.

Lemma 2.6]is encoded through the definition of four mutually recursive functions: as the statement
involves transitions via a generic action , the first two prove the result for free transitions, while the
last two demonstrate the result for bound transitions; moreover, since the proof is carried out by con-
currently establishing two symmetrical assertions, the odd functions prove the left-to-right statement,
while the even ones demonstrate the right-to-left statement. We present the signature of the first function
cong_fstepleft_impl_stepright, which proves the left-to-right assertion for free transitions:

rec cong_fstepleft_impl_fstepright: (g:ctx) [ghF P cong Q1 — [g - fstep P A P’]
— [g F ex_fstepcong P Q A P’] = ...

Mirroring the informal proof, this lemma is proved by a long induction on the structure of the given
congruence; in most of the subcases, case analysis of the given transition is performed as well.
A final ingredient for the proof of Theorem [2]is the auxiliary lemma:

rec red_rew_impl_fstepcong: (g:ctx) [g - ex_red_rew P Q]
— [g F ex_fstepcong P P f_tau Q] = ...

The proof proceeds by induction on the structure of the given object of type [g - ex_red_rew P Q];
in both the base case and the inductive case, a key factor consists in the application of the function
cong_fstepright_impl_fstepleft.

We are ready to prove Theorem

rec red_impl_fstepcong: (g:ctx) [gF P red Q1 — [g F ex_fstepcong P P f_tau Q] =
/ total r (red_impl_fstepcong _ _ _ r) /
fn r = let [g F D1] = red_impl_red_rew r in

let [g F D2] = red_rew_impl_fstepcong [g F D1] in [g + D2]

Given r representing the reduction P — Q, we apply the function red_impl_red_rew to it
returning an object D1 of type [g F ex_red_rew P Q] that encodes the following congruences:
P=(vwi)-- (vwn)((Xy.Ry | x(2).R2) | S) and Q = (vwy) -+ (vwa)((R1 | R2{y/z}) | S), for some Ry, Ry,
Sand wy,...,w,. Finally, we invoke the auxiliary function red_rew_impl_fstepcong, which unfolds the
argument D1 and returns the desired object of type [g F ex_fstepcong P P f_tau QJ.

4 Evaluation and Conclusions

The reader may wonder “Is that it?” We sympathize with the feeling: what is remarkable in this for-
malization is how (mostly) uneventful it has been. Once we had settled on using (weak) HOAS and a
specialized proof environment such as Beluga — which, given our lineage, was not much of a stretch —
the Honsell/Miller encoding of the labelled transition system removed all issues related to scope extru-
sion and Beluga’s remarkable conciseness did the rest, turning 30 pages of ISIEX proofs, which still skip
many steps, into some 700 lines of proof terms.

Remarkably, the structure of the formal proof closely mirrors the informal one: having eliminated
the 7 technical lemmas thanks to the HOAS encoding, both proofs share the same 6 lemmas, proved in
the same fashion. Some parts of the formal proof are covered by 22 additional lemmas which deal with
the unfolding of the existential types while another 4 lemmas result from the mutual recursion induced
by the encoding.

In our biased opinion, this uneventfulness does not trivialize the accomplishment: we have provided
a compact and elegant solution of a benchmark problem, which, after all, is supposed to be challenging:
the fact that this has not been a heroic feat is a testament to the merits of the HOAS encoding and to
the long line of research that has made meta-level reasoning over HOAS specifications possible. It also
suggests that, once we put on the HOAS “spectacles”, the binders of the m-calculus are not that different
from those of the A-calculus, in the sense that the meta-level binder will gladly model scope extrusion,
under the right encoding.

3Given how widespread existential (and conjunctive) statements are in this development, it would be helpful if Beluga could
provide some syntactic sugar, similarly to Agda, and a way to unfold those definitions automatically.

Beluga has shown to be a reliable system: we did stress the termination checker, with a heavy use
of mutually defined recursive functions. We managed to get around the current lack of support for
lexicographic induction; our technique could have broader applicability, for instance in verifying totality
results such as the admissibility of cut elimination that rely on nested induction for their termination
proofs [16]. We also established coverage, again getting around the minor glitch that we have mentioned
above in the proof of the right-to-left direction of the Harmony Lemma.

Since the benchmark is amenable to a weak HOAS encoding, this begs the question of why not
pursue its solution in a general proof assistant such as Coq. While weak HOAS is indeed consistent with
monotonic inductive types, it is well known that the full dependent function space of a theory such as the
Calculus of Constructions is incompatible with the intensional quantification featured by LF-like type
theories. Workarounds exist: the most successful one is the “Theory of Contexts” [10]], where additional
predicates concerning freshness and non-occurrence of names are added to specifications such as of the
LTS. Further, ToC assumes some axioms regulating the properties of names and abstraction over names
(i.e. “contexts”) in order to reify what LF-like frameworks provide natively. To be fair, it is unclear to us
which role ToC would play in a Coq solution of CCFB.2, but for the rest of the Concurrent Benchmark,
the outcome is not pretty (believe us, we tried). Of course, there is no obstacle in abandoning HOAS for
concrete encodings and we are looking forward to comparing such solutions to ours.

Although CCFB.2 does not ask for it, we conjecture that it would be easy, albeit tedious, to extend our
solution to account for other features of the m-calculus, namely sums, replication and match. Mismatch,
which is handled in [10]], is rather problematic in HOAS, since the systems that support it have no native
notion of negation.

The adoption of Beluga as a proof environment for this formalization is motivated by our endeavor
(together with Pientka’s group) to give an overall HOAS solution to all the challenges listed in CCFB,
which include type safety for (linear) session types and turning strong barbed bisimilarity into a congru-
ence. For this, Beluga is a strong candidate: in fact the type safety challenge is already in the bag, thanks
to the techniques developed in [20]. The coinduction part is more challenging, but we have a good track
record in a similar benchmark [[14]. Solving the rest of CCFB in Beluga may also shed some light on the
role of the V quantifier [[13] as a meta-reasoning tool, compared to Beluga’s use of contextual LF as a
specification language.

A Appendix: Late vs Early Transitions

An alternative to the late semantics presented in the paper is the early semantics. As its name suggests,
it is characterized by the fact that substitutions of names received in interaction are performed as soon
as possible, namely during the execution of the S-IN rule. The syntax of actions is the same as in the
late semantics; however, the name y occurring in an input action x(y) is considered to be free. As for

transitions, now denoted as i the rules S-IN, the two S-CoM rules and the two S-CLOSE are replaced
by those in Fig. [7] (“right” rules are omitted for brevity). Note how the S-IN rule now exhibits the
substitution of the input name; conversely, in the S-CoM rules, both of the names x and y occurring in
the actions of the given transitions must coincide and no substitution takes place in the conclusion.

S-CoM-L S-CLOSE-L
S-IN P2p oWy PP 0™ o ¢
22).P 2 ply/2) PlQ 5 P|Q PlQ % (vo)(P'| Q)

Figure 7: Early transition semantics rules.

In order to prove the equivalence of the two semantics, we follow Parrow’s approach in [[15]: namely,
our objective is to demonstrate that the two semantics allow to infer the same 7-transitions. Both direc-
tions are achieved by induction on the depth of the inference of the given transitions and rely on some
additional lemmas, which state a correspondence between input/output transitions of the two semantics
as well. The only non-trivial correspondence lies between input (in the late semantics) and free input (in
the early semantics), which is defined as follows:

P Q iff there are Q" and w such that P ), Q' and Q = Q'{y/w}. @)

We begin the Beluga formalization by encoding the two semantics in the same environment (Fig. [8)).
The type f_act presents a new constructor f_in for free input actions in the early semantics; as for
transitions, the constructors expressing identical rules in the two semantics are omitted for brevity. We
observe that the ebs_in constructor representing bound input transitions in the early semantics cannot be
eliminated, since bound input transitions are needed as premises in the rules introduced by the efs_close
constructors. For consistency of notation, the types fstep and bstep for late transitions have been
renamed as late_fstep and late_bstep.

The next ingredient for the formalization of the semantics equivalence is the definition of the type
family ex_latebs, which encodes the existence of a late transition such as in the correspondence (2):

LF ex_latebs: proc — names — names — proc — type =
| 1bs: late_bstep P (b_in X) \w.(Q’> w) — eqp Q (@’ Y) — ex_latebs P X Y Q

% Free Actions % Bound Actions

LF f_act: type = LF b_act: type =
| f_in: names — names — f_act | b_in: names — b_act
| f_out: names — names — f_act | b_out: names — b_act

| f_tau: f_act

% Early Transition Relation
LF early_fstep: proc — f_act — proc — type =
| efs_in: early_fstep (p_in X P) (f_in X Y) (P Y)
| efs_coml: early_ fstep P (f_out X Y) P’ — early_fstep Q (f_in X V) @’
— early_fstep (P p_par Q) f_tau (P’ p_par Q’)
| efs_com2: early_fstep P (f_in X Y) P’ — early_fstep Q (f_out X Y) Q°
— early_fstep (P p_par Q) f_tau (P’ p_par Q’)
| efs_closel: early_bstep P (b_out X) P’ — early_bstep Q (b_in X) Q’
— early_fstep (P p_par Q) f_tau (p_res \z.((P’ z) p_par (Q’ z2)))
| efs_close2: early_bstep P (b_in X) P’ — early_bstep Q (b_out X) Q’
— early_fstep (P p_par Q) f_tau (p_res \z.((P’ 2z) p_par (Q’ 2)))

and early_bstep: proc — b_act — (names — proc) — type =
| ebs_in: early_bstep (p_in X P) (b_in X) P

Figure 8: Encoding of actions and early transition.

We then list the signature of the functions finp_earlytolate and finp_latetoearly, encoding the
correspondence between free input transitions in the early semantics and input transitions in the late
semantics. The correspondence between the other types of actions is performed analogously with two
functions for each case; since their formalization is straightforward, it is omitted.

rec finp_earlytolate: (g:ctx) [g - early_fstep P (f_in X Y) Q]
— [g F ex_latebs PX Y Q] = ...

rec finp_latetoearly: (g:ctx) {Y:[g - names]} [g F ex_latebs P X Y Q]
— [g F early_fstep P (f_in X Y) Q] = ...

In the second statement, the input name Y needs to be passed as an explicit argument, otherwise Beluga
would not be able to reconstruct it during some further calls of this lemma. The proofs of both lemmas
are straightforward inductions on the given derivation.

We can now state the signature of the functions tau_earlytolate and tau_latetoearly, which
encode the equivalence of the two semantics:

rec tau_earlytolate: (g:ctx) [g - early_fstep P f_tau Q]
— [g F late_fstep P f_tau Q] = ...

rec tau_latetoearly: (g:ctx) [g F late_fstep P f_tau Q]
— [g - early_fstep P f_tau Q] = ...

The proof follows by induction on the given transition. In case the transition is obtained through a S-Com
or S-CLOSE rule, we apply the previously defined lemmas in order to turn an early input/output transition
into a corresponding late input/output transition and then conclude.

